Tutorial 7:  Loop Counters and Repeated Operations

New Instructions Introduced

BANZ

NOP

RPT

RPTZ

Overview of Tutorial

A commonly encountered task within a program is to execute an operation multiple times. Users of high level languages will be familiar with the way this is implemented using the FOR, WHILE and REPEAT-UNTIL (DO-WHILE) constructions. This tutorial shows the usage of some of the special TMS320C5000 instructions designed for controlling loops. 

Implementing FOR and WHILE loops

A typical way to implement a control loop is to use a counter that starts at a preset value and is incremented every time the body of the loop is executed. When the counter reaches a certain value, execution of the loop terminates.  

We can implement a control loop in the C language using a FOR loop:

Example 7-1.

	
	

	unsigned int i, j;
	 

	for ( i = 0 ; i < 20 ; i++)
	// 20 iterations.

	{ 
	

	 j++;  
	// Some operation.

	}
	


The code in Example 7-1 can also be written in terms of a WHILE loop:

Example 7-2.

	
	

	unsigned int i, j;
	 

	i = 0; 
	

	while ( i < 20 )
	// 20 iterations.

	{ 
	

	 i++;
	// increment counter.

	 j++;  
	// Some operation.

	}
	


We can implement a FOR or a WHILE loop in Examples 7-1 and 7-2 using the accumulator for variable i and the instruction BC (branch conditionally). Assuming that variable j is stored at data memory address 70h:

Example 7-3.

	
	
	

	
	LD #0, DP
	; Page 0. Gain access to data

; memory addresses 60h to 7Fh.

	
	LD #0, A
	; Initialize variable i to 0.

	loop1:
	SUB #20, A
	; Accumulator A - 20.

	
	BC done1, AGEQ 
	; Test if variable i is greater

; than or equal to 20.

	
	ADD #20, A
	; Restore accumulator A to value 

; it was before the subtraction.

	
	ADD #1, A
	; Increment variable i.

	
	ADDM #1, 70h
	; Increment variable j.

	
	B loop1
	; Go round again.

	done1:
	
	


Performing a comparison of the counter i with the terminating value, here 20, is not the most efficient of operations. 

Implementing a REPEAT-UNTIL (DO-WHILE) Loop

The other type of loop we can use is the REPEAT-UNTIL type of loop, known as the DO-WHILE loop in C. Example 7-4 shows the implementation of a DO-WHILE loop using a counter that counts down and terminates when the count reaches zero:

Example 7-4.

	
	

	unsigned int i = 20;
	// Assign variable.  

	do 
	// Start of loop.

	{
	

	 i--; 
	// Decrement counter.

	 j++;  
	// Some operation in loop.

	} while ( i > 0 )
	// Test condition. 


To implement Example 7-4 in TMS320C54x assembly language, we assign variable i to accumulator A and variable j to a data memory address, in this case data memory address 70h:

Example 7-5.

	
	
	

	
	LD #0, DP
	; Page 0. Gain access to data

; memory addresses 60h to 7Fh.

	
	LD #20, A
	; Initialize variable i to 20.

	loop2:
	SUB #1, A
	; Decrement variable i.

	
	ADDM #1, 70h   
	; Increment variable j.

	
	BC loop2, ANEQ
	; If variable i is not equal

; to zero, then branch to the

; label loop2, otherwise

; terminate loop by dropping

; through to the next line.  


If a comparison is made of Examples 7-3 and 7-5, it will be seen that as far as code efficiency is concerned, the DO-WHILE in Example 7-5 loop requires less instructions.

Implementing a Loop Counter using the TMS320C5000

Another way to write a DO-WHILE loop to carry out 20 operations is shown in Example 7-6:

Example 7-6.

	
	

	unsigned int i = 19;
	// Assign variable.  

	do 
	// Start of loop.

	{
	

	 j++;  
	// Some operation in loop.

	} while ( i-- != 0 )
	// Decrement and test. 


In order to carry out 20 operations we need to load the counter i with 20 - 1 = 19 because the counter is decremented after the test.

The special branch instruction BANZ (branch on auxiliary register non-zero) is provided for control loops and uses one of the auxiliary registers AR0 to AR7. This leaves accumulators A and B free for other purposes.  

If we assign variable i to auxiliary register AR3, then we can implement the DO-WHILE loop from Example 7-6 using the instruction BANZ (branch on auxiliary register non-zero):

Example 7-7.

	
	
	

	
	LD #0, DP
	; Page 0. Gain access to data

; memory addresses 60h to 7Fh.

	
	LD #19, AR3
	; Start with 20 - 1 = 19 in

; AR3.  

	loop3:
	ADDM #1, 70h
	; Increment variable j.

	
	BANZ loop3, *AR3-
	; If the value in auxiliary

; register AR3 is non-zero,

; then branch to the label

; loop3. Otherwise, drop

; through to the next line

; of the program. In either

; case, decrement AR3. 


The instruction BANZ takes two operands. The first operand is the label representing the address in program memory to which the program execution is redirected. The second operand is the auxiliary register to be tested. In this case, the symbol * does not indicate indirect addressing; it simply means that the operation is carried out on an auxiliary register. It is important to put one less than the desired number of counts in the auxiliary register to be tested.    

In Example 7-7, the instruction BANZ tests the auxiliary register AR3, and if the contents are non-zero, branches (jumps) to the label loop3. If the contents of the current auxiliary register AR3 are zero, then the program execution continues linearly to the next line. Whether or not a branch occurs, the minus sign in the operator *AR3- decrements auxiliary register AR3. 

Branches to the Same Address

Should we wish the program to branch to the current address we can write:

Example 7-8.

	
	

	BANZ $, *AR3-
	; Branch to current address until the

; value in AR3 is decremented to zero. 


So that the loop terminates, it is necessary to either increment or decrement the auxiliary register used as the second operand. Example 7-9 does not alter the auxiliary register and therefore will remain in the loop forever: 

Example 7-9.

	
	

	BANZ $, *AR5
	; Auxiliary register AR5 is neither

; incremented nor decremented. This

; loop will never terminate. 


The following are incorrect syntax and are not allowed:

Example 7-10.

	
	

	BANZ loop2
	; An auxiliary register is required as

; the second operand.

	BANZ loop2, AR4
	; Missing symbol * before the name of

; auxiliary register.

	BANZ *AR4-
	; Missing label for the first operand.


Incrementing the Count

In the usage of the instruction BANZ to date, we have chosen to decrement the auxiliary register. We also have the possibility of incrementing the auxiliary register:

Example 7-11.

	
	
	

	
	STM #0FFF7h, AR3
	; Start with -9 in AR3.  

	label3:
	BANZ label3, *AR3+
	; If the value in the

; auxiliary register (AR3)

; is non-zero, then branch

; to the label label3.

; Otherwise, execute the

; next line of the program.

; In either case, increment

; AR3. 


It is more common to use the *AR3- variation to count down rather than to count up. The desired number of counts can then be put directly into the current auxiliary register. When counting up, a negative number must be put into the current auxiliary register which may be more difficult to read and debug.

Note that the current auxiliary register is modified every time the instruction BANZ is executed, whether or not there is a branch. When the loop terminates, the current auxiliary register will no longer contain zero. 

Working in Non-Unit Values

So far, every time we have executed the body of the loop, we have always changed the count by +1 or -1. We may wish to use a counter that works in some other units, for example 10. Consider the following example of C code that counts down from 100 in units of 10:

Example 7-12.

	
	

	unsigned int i = 100;
	// Initialise counter to 100. 

	while ( i != 0) 
	// Loop while counter is not zero.

	{
	

	 i -= 10;
	// Decrement counter by 10.

	}
	


The equivalent to Example 7-12 in TMS320C5000 assembly language would be:

Example 7-13.

	
	

	STM #10, AR0
	; Load auxiliary register AR0 with the

; amount that the loop counter is to be

; decremented by, in this case 10.

	STM #100, AR3
	; Load counter AR3 with 100.

	BANZ $, *AR3-0
	; Test the auxiliary register (AR3). If

; non-zero, branch to the same address, 

; otherwise execute the next line of 

; code. In both cases, subtract the

; value in AR0 from the auxiliary

; register (AR3).  


Here the operand *AR3-0 means decrement the auxiliary register AR3 by the value in register AR0. The value in AR3 will be decremented by units of 10. The first time through, the auxiliary register AR3 contains 100 decimal. After the instruction BANZ, it will then contain AR3 - AR0 = 100 - 10 = 90. The instruction BANZ will be executed a total of 10 times before dropping through to the next line.

In a similar way, we can also count up in units other than 1, as shown in Example 7-14. 

Example 7-14.

	
	

	STM #10, AR0
	; Load auxiliary register AR0 with the

; amount that the loop counter AR3 is to

; be incremented. 

	STM #0FFA6h, AR3
	; Load counter AR3 with -90.

	BANZ $, *AR3+0
	; Test the auxiliary register (AR3). If

; non-zero, branch to the same address, 

; otherwise drop through to the next

; line. In either case, add the

; value in AR0 to auxiliary register

; AR3.  


Note that the instruction BANZ tests for zero, so that if the count does not go exactly to zero, the loop will not terminate as expected. This is shown in Example 7-15.

Example 7-15.

	
	

	STM #3, AR0
	; Load auxiliary register AR0 with the

; amount that the loop counter AR2 is to

; be decremented by, in this case 3.

	STM #11, AR2
	; Load counter AR2 with 11.

	BANZ $, *AR2-0
	; Test the auxiliary register AR2. If

; non-zero, branch to the same address, 

; otherwise execute the next line. In

; both cases, subtract the value in AR0

; from the auxiliary register AR3.  


In Example 7-15, the count in AR2 will be 11, 8, 5, 2, -1, -4 etc. and hence will not reach zero when expected. This means that the number of iterations performed will be greater than that intended.

Filling Data Memory Addresses

So far we have seen how to use the instruction BANZ for loop counters. Let us now look at a practical example. Say we wish to load the 16 successive data memory addresses starting at 300h with AAAAh, for example as part of a RAM test. This is shown in the following C code:  

Example 7-16.

	
	

	unsigned int *AR6;
	// Pointer to memory. 

	unsigned int AR5 = 16;
	// 16 iterations.

	AR6 = 0x300h;
	// First memory address.  

	do
	// Count down until AR5 is

// zero.

	{
	

	 *AR6 = 0xAAAA; 
	// Load memory location.

	  AR6++;
	// Point to next address.

	} while (AR5-- != 0);
	


Because we are going to work on a series of different data memory addresses, it is easier to use indirect addressing than direct addressing. We shall use AR5 as our loop counter and AR6 as the pointer to the data. This is implemented in Example 7-17 and illustrates how to use multiple auxiliary registers in the same code fragment.

Example 7-17.

	
	
	

	
	STM #300h, AR6
	; Use auxiliary register AR6 as

; a pointer to the start of

; data at data memory address

; 300h. 

	
	STM #16-1, AR5 
	; Use auxiliary register AR5 as

; the loop counter. Operation

; to be executed 16 times.  

	loop4:
	ST #0AAAAh, *AR6+
	; Save AAAAh at the data

; memory address pointed

; to by AR6. Point to the next

; higher address.

	
	BANZ loop4, *AR5-
	; If AR5 contains zero then

; terminate, otherwise branch

; to the label loop4. In

; either case, decrement AR5.


Note that the operand *AR6+ increments the address contained in auxiliary register AR6, not the contents of the data memory address pointed to by AR6. After the code in Example 7-17 has been executed, the data memory addresses 300h to 310h will each contain the value AAAAh.

Repeated Operations

Say we wish to generate a short time delay, for example to allow an input to settle before taking a reading. We need to execute some instructions a fixed number of times that have no effect upon the program execution. Ideal for this purpose is the instruction NOP (no operation).

Example 7-18.

	
	

	NOP
	; No operation. This instruction has no

; effect except to take up execution time.


To make up a delay might require the instruction NOP to be written out 100 times. This is wasteful use of program memory. Instead we combine the instruction NOP with the instruction RPT (repeat next instruction).

Example 7-19. 

	
	

	RPT #99
	; Repeat the next instruction 99 times. 

	NOP
	; No operation.


The instruction RPT takes one operand which is the number of repeats. This is where it is easy to make a mistake. The constant gives the number of repeats, not the total number of times the instruction following the instruction RPT is executed. The above code in fact causes the instruction NOP to be executed a total of 100 times. It is easy to have an “out by one” error. With this in mind, we might write an operation to execute the instruction NOP a total of 1000 times as:

Example 7-20.

	
	

	RPT #(1000-1)
	; Repeat the next instruction 999 times. 

	NOP
	; No operation.


The instruction RPT is a very effective way of causing an instruction to be executed many times while writing a minimum amount of code. There is however, a limitation. Only one instruction following the instruction RPT will be repeated. Should it be necessary to repeat a block of code, then we would use the instruction BANZ. 

Not all instructions can be used with the instruction RPT. Most instructions that carry out simple operations such as load and store can be repeated. However, instructions that change the program counter, such as the instructions B (branch unconditionally) and BANZ are not repeatable. 

Using Repeats to Fill Data Words in Memory.

Consider the following task. We wish to fill data memory addresses 300h to 3FFh with the value 0000h. In a practical situation, this may be used as part of a start-up routine to fill the data memory with known values.

To use the analogy of C code, we could write:

Example 7-21.

	
	

	unsigned int i;
	// Iterations counter.

	unsigned int *AR3 ;
	// Pointer to memory.

	AR3 = 300h ;
	// Point to start address.

	for (i = 0 ; i < 255 ; i++)
	// Perform action 255 times.

	{ *AR3++ = 0x00;  }
	// Fill memory with 00h. 


In order to write to successive addresses we need to use indirect addressing combined with the instruction RPT, as shown in Example 7-22:

Example 7-22.

	
	

	STM #300h, AR3
	; Store address of start of data memory

; block in AR3. AR3 = 300h.

	LD  #0, A
	; Clear accumulator A.

	RPT #(255-1)
	; Total 255 operations.

	STL A, *AR3+
	; Load low word of accumulator A (value =

; 0) into the data memory address

; pointed to by AR3. Increment value in

; AR3 to point to the next data memory

; address.


We initially set up auxiliary register AR3 as a pointer to the first data memory address we wish to load. There are 255 locations to fill with 0 so we require 255 - 1 = 254 repeats. After we load the low word of the accumulator into the data memory address, we also increment AR3 to point to the next data memory address. The important point here is that indirect addressing that makes it possible to load successive data memory locations with a single instruction STL.

We can make a code reduction to Example 7-22. The instructions to clear an accumulator and to repeat the following instruction can be condensed into the single instruction RPTZ (repeat next instruction and clear accumulator).   

Example 7-23.

	
	

	STM #300h, AR3
	; Store the address of the start of data

; memory block in auxiliary register

; AR3. AR3 = 300h.

	RPTZ A, #(255-1)
	; Clear accumulator A then repeat

; following operation 254 times.

	STL A, *AR3+
	; Load low word of accumulator A (value

; = 0) into the data memory address

; pointed to by AR3. Increment the value ; in AR3 to point to the next data

; memory address.


Unlike the instruction RPT, the instruction RPTZ (repeat next instruction and clear accumulator) takes two operands. The first operand is the accumulator to be cleared and the second operand specifies the number of repeats. 

Note that the instruction RPTZ (repeat next instruction and clear accumulator) clears the accumulator once only, and not during every repeat. We can also apply the instruction RPTZ to accumulator B: 

Example 7-24.

	
	

	RPTZ B, #(10-1)
	; Clear accumulator B then repeat

; following operation 9 times.

	ADD #1, B
	; Increment accumulator B each repeat.

; When the repeat terminates,

; accumulator B will contain 9, not 1. 


Upgrading from the TMS320C2000 to the TMS320C5000

The instruction BANZ (branch on auxiliary register non-zero) is supported by the TMS320C2000 and the TMS320C5000. However, the way the instruction is used does differ. 

On the TMS320C5000, the instruction BANZ always takes two operands, whereas the earlier devices were able to use between one and three. The differences between the various devices are illustrated in Table 7-1:    

Table 7-1. Comparison of BANZ Instructions

	Description
	 TMS320C2000 

 Instruction
	TMS320C5000

 Instruction(s)

	Loop using decrement of auxiliary register.
	BANZ loop5
	BANZ loop5, *ARx-

	Loop using decrement of auxiliary register and set new ARP.
	BANZ loop6, *-, AR2
	BANZ loop6, *ARx-

LD #2, ARP

	Loop using increment of auxiliary register.
	BANZ loop7, *+
	BANZ loop7, *ARx+

	Subtract AR0 from auxiliary register. 
	BANZ loop8, *0-
	BANZ loop8, *ARx-0

	Add AR0 to auxiliary register.
	BANZ loop9, *0+
	BANZ loop9, *ARx+0


ARx = One of auxiliary registers AR0 to AR7.

On the TMS320C2000, if no second operand is entered, it is taken to be the default operand of *-.  On the TMS320C5000, the second operand must always be entered.

Unlike the TMS320C2000, the TMS320C5000 does not allow the current auxiliary register (ARP) to be changed as part of the instruction.   

The repeat instructions are used in a similar way across the two families, but there is a small difference in syntax. In order to implement the equivalent of the TMS320C5000 instruction RPTZ on the TMS320C2000, two instructions are required. 

Table 7-2.  Comparison of Repeat Instructions

	Description
	 TMS320C2000

Instruction(s)
	TMS320C5000

 Instruction

	Repeat following command 10 times.
	RPT #10
	RPT #10

	Clear accumulator and then repeat following instruction 99 times.
	LACL #0

RPT #99
	RPTZ Acc, #99


Here Acc = Accumulator A or accumulator B.

Questions

	1.
	Why might we use a DO-WHILE construct in preference to a FOR construct?

	2.
	Add comments to the following code fragment:
loop1:  STL A, *AR2+

        BANZ loop1, *AR3-

	3.
	If using the instruction BANZ for a loop counter, why would we tend not use AR0 for our iterations counter?

	4.
	How many times will the following loop execute?

       STM #19, AR5

loop3: STL  B, *AR6-

       BANZ loop3, *AR5- 

	5.
	What will be the value in auxiliary register AR3 when the line after the instruction BANZ is reached?

STM #10, AR3

BANZ $, *AR3- 

	6.
	What is meant by the expression LD A, *AR6+0?

	7.
	What is meant by the expression STL B,*AR2-0?

	8.
	How do we use the instruction NOP?

	9.
	It is intended to do 9 repeats using the following code:

RPT 9

BANZ $, *AR5-

When debugging, the instruction BANZ is not repeated. Why is this? 

	10.
	It is intended to execute the instruction STL a total of 100 times using the following code:

RPT 100

STL A, *AR2+
When debugging, the instruction STL is executed 101 times. Why is this? 

	11.
	How many times is accumulator B cleared to zero in the following example:

RPTZ B, 100

	12.
	There are cases where we cannot use the instruction RPT and have to use the instruction BANZ instead. Why is this?

